

# Formalization of Refinement Calculus for Reactive Systems

Viorel Preoteasa  
Aalto University, Finland

October 11, 2017

## Abstract

We present a formalization of refinement calculus for reactive systems. Refinement calculus is based on monotonic predicate transformers (monotonic functions from sets of post-states to sets of pre-states), and it is a powerful formalism for reasoning about imperative programs. We model reactive systems as monotonic property transformers that transform sets of output infinite sequences into sets of input infinite sequences. Within this semantics we can model refinement of reactive systems, (unbounded) angelic and demonic nondeterminism, sequential composition, and other semantic properties. We can model systems that may fail for some inputs, and we can model compatibility of systems. We can specify systems that have liveness properties using linear temporal logic, and we can refine system specifications into systems based on symbolic transitions systems, suitable for implementations.

## Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Linear Temporal Logic</b>	<b>2</b>
<b>3</b>	<b>Monotonic Predicate Transformers</b>	<b>6</b>
3.1	Basic predicate transformers . . . . .	6
3.2	Conjunctive predicate transformers . . . . .	8
3.3	Fusion of predicate transformers . . . . .	10
<b>4</b>	<b>Reactive Systems</b>	<b>11</b>
4.1	Symbolic transition systems . . . . .	12
4.2	Example: COUNTER . . . . .	17
4.3	Example: LIVE . . . . .	18

# 1 Introduction

This is a formalization of refinement calculus for reactive systems that is presented in [6].

Refinement calculus [1, 3] has been developed originally for input output imperative programs, and is based on a predicate transformer semantics of programs with a weakest precondition interpretation.

We extend the standard refinement calculus to reactive systems [4]. Within our framework a reactive system is seen as a system that accepts as input an infinite sequence of values and produces as output an infinite sequence of values. The semantics of these systems is given as *monotonic property transformers*. These are monotonic functions which maps sets of output sequences (output properties) into sets of input sequences (input properties). For a set of output sequences  $q$ , the monotonic property transformer  $S$  applied to  $q$  returns all input sequences from which the computation of  $S$  always produces a sequence from  $q$ .

Our work extends also the relational interfaces framework of [7] which can handle only finite safety properties to infinite properties and liveness.

This formalization is organized in three sections. Section 2 presents an algebraic formalization of linear temporal logic. Section 3 introduces basic constructs from refinement calculus, and finally Section 4 applies the refinement calculus to reactive systems.

```
theory Temporal imports Main  
begin
```

## 2 Linear Temporal Logic

In this section we introduce an algebraic axiomatization of Linear Temporal Logic (LTL). We model LTL formulas semantically as predicates on traces. For example the LTL formula  $\alpha = \Box \Diamond (x = 1)$  is modeled as a predicate  $\alpha : (nat \Rightarrow nat) \Rightarrow bool$ , where  $\alpha x = True$  if  $x i = 1$  for infinitely many  $i : nat$ . In this formula  $\Box$  and  $\Diamond$  denote the always and eventually operators, respectively. Formulas with multiple variables are modeled similarly. For example a formula  $\alpha$  in two variables is modeled as  $\alpha : (nat \Rightarrow 'a) \Rightarrow (nat \Rightarrow 'b) \Rightarrow bool$ , and for example  $(\Box \alpha) x y$  is defined as  $(\forall i. \alpha x[i..] y[i..])$ , where  $x[i..] j = x (i + j)$ . We would like to construct an algebraic structure (Isabelle class) which has the temporal operators as operations, and which has instantiations to  $(nat \Rightarrow 'a) \Rightarrow bool$ ,  $(nat \Rightarrow 'a) \Rightarrow (nat \Rightarrow 'b) \Rightarrow bool$ , and so on. Ideally our structure should be such that if we have this structure on a type  $'a :: temporal$ , then we could extend it to  $(nat \Rightarrow 'b) \Rightarrow 'a$  in a way similar to the way Boolean algebras are extended from a type  $'a :: boolean\_algebra$  to  $'b \Rightarrow 'a$ . Unfortunately, if we use for example  $\Box$  as primitive operation on our temporal structure, then we cannot extend  $\Box$

from  $'a :: \text{temporal}$  to  $(\text{nat} \Rightarrow 'b) \Rightarrow 'a$ . A possible extension of  $\square$  could be

$$(\square \alpha) x = \bigwedge_{i:\text{nat}} \square(\alpha x[i..]) \text{ and } \square b = b$$

where  $\alpha : (\text{nat} \Rightarrow 'b) \Rightarrow 'a$  and  $b : \text{bool}$ . However, if we apply this definition to  $\alpha : (\text{nat} \Rightarrow 'a) \Rightarrow (\text{nat} \Rightarrow 'b) \Rightarrow \text{bool}$ , then we get

$$(\square \alpha) x y = (\forall i j. \alpha x[i..] y[j..])$$

which is not correct.

To overcome this problem we introduce as a primitive operation  $!! : 'a \Rightarrow \text{nat} \Rightarrow 'a$ , where  $'a$  is the type of temporal formulas, and  $\alpha!!i$  is the formula  $\alpha$  at time point  $i$ . If  $\alpha$  is a formula in two variables as before, then

$$(\alpha!!i) x y = \alpha x[i..] y[i..].$$

and we define for example the the operator always by

$$\square \alpha = \bigwedge_{i:\text{nat}} \alpha!!i$$

#### notation

*bot* ( $\perp$ ) **and**  
*top* ( $\top$ ) **and**  
*inf* (**infixl**  $\sqcap$  70)  
**and** *sup* (**infixl**  $\sqcup$  65)

**class** *temporal* = *complete-boolean-algebra* +

**fixes** *at* ::  $'a \Rightarrow \text{nat} \Rightarrow 'a$  (**infixl**  $!!$  150)  
**assumes** [*simp*]:  $a !! i !! j = a !! (i + j)$   
**assumes** [*simp*]:  $a !! 0 = a$   
**assumes** [*simp*]:  $\top !! i = \top$   
**assumes** [*simp*]:  $\neg(a !! i) = (\neg a) !! i$   
**assumes** [*simp*]:  $(a \sqcap b) !! i = (a !! i) \sqcap (b !! i)$

**begin**

**definition** *always* ::  $'a \Rightarrow 'a$  ( $\square$  (-) [900] 900) **where**  
 $\square p = (\text{INF } i . p !! i)$

**definition** *eventually* ::  $'a \Rightarrow 'a$  ( $\diamond$  (-) [900] 900) **where**  
 $\diamond p = (\text{SUP } i . p !! i)$

**definition** *next* ::  $'a \Rightarrow 'a$  ( $\circ$  (-) [900] 900) **where**  
 $\circ p = p !! (\text{Suc } 0)$

**definition** *until* ::  $'a \Rightarrow 'a \Rightarrow 'a$  (**infix** *until* 65) **where**  
 $(p \text{ until } q) = (\text{SUP } n . (\text{INFIMUM } \{i . i < n\} (at p))) \sqcap (q !! n)$   
**end**

Next lemma, in the context of complete boolean algebras, will be used to prove  $-(p \text{ until } -p) = \Box p$ .

```

context complete-boolean-algebra
begin
  lemma until-always: ( $INF\ n. (SUP\ i : \{i. i < n\} . - p\ i) \sqcup ((p :: nat \Rightarrow 'a)\ n) \leq p\ n$ )
    <proof>

end

```

We prove now a number of results of the temporal class.

```

context temporal
begin
  lemma [simp]: ( $a \sqcup b \ !!\ i = (a \ !!\ i) \sqcup (b \ !!\ i)$ )
    <proof>

  lemma always-less [simp]:  $\Box p \leq p$ 
    <proof>

  lemma always-and:  $\Box (p \sqcap q) = (\Box p) \sqcap (\Box q)$ 
    <proof>

  lemma eventually-or:  $\Diamond (p \sqcup q) = (\Diamond p) \sqcup (\Diamond q)$ 
    <proof>

  lemma neg-until-always:  $-(p \text{ until } -p) = \Box p$ 
    <proof>

  lemma neg-always-eventually:  $\Box p = - \Diamond (- p)$ 
    <proof>

  lemma neg-true-until-always:  $-(\top \text{ until } -p) = \Box p$ 
    <proof>

  lemma true-until-eventually:  $(\top \text{ until } p) = \Diamond p$ 
    <proof>
end

```

Boolean algebras with  $b!!i = b$  form a temporal class.

```

instantiation bool :: temporal
begin
  definition at-bool-def [simp]: ( $p::bool \ !!\ i = p$ )
  instance <proof>
end

```

```

type-synonym 'a trace = nat  $\Rightarrow$  'a

```

Assuming that  $'a :: temporal$  is a type of class *temporal*, and  $'b$  is an arbitrary type, we would like to create the instantiation of  $'b \text{ trace} \Rightarrow 'a$

as a temporal class. However Isabelle allows only instantiations of functions from a class to another class. To solve this problem we introduce a new class called *trace* with an operation  $suffix :: 'a \Rightarrow nat \Rightarrow 'a$  where  $suffix\ a\ i\ j = (a[i..])\ j = a\ (i + j)$  when  $a$  is a trace with elements of some type  $'b$  ( $'a = nat \Rightarrow 'b$ ).

```

class trace =
  fixes suffix :: 'a  $\Rightarrow$  nat  $\Rightarrow$  'a (-[- ..] [80, 15] 80)
  assumes [simp]: a[i..][j..] = a[i + j..]
  assumes [simp]: a[0..] = a
begin
  definition next-trace :: 'a  $\Rightarrow$  'a ( $\odot$  (-) [900] 900) where
     $\odot\ p = p[Suc\ 0..]$ 
end

instantiation fun :: (trace, temporal) temporal
begin
  definition at-fun-def: (P:: 'a  $\Rightarrow$  'b) !! i = ( $\lambda\ x . (P\ (x[i..]))$ ) !! i
  instance <proof>
end

```

In the last part of our formalization, we need to instantiate the functions from *nat* to some arbitrary type  $'a$  as a trace class. However, this again is not possible using the instantiation mechanism of Isabelle. We solve this problem by creating another class called *nat*, and then we instantiate the functions from  $'a :: nat$  to  $'b$  as traces. The class *nat* is defined such that if we have a type  $'a :: nat$ , then  $'a$  is isomorphic to the type *nat*.

```

class nat = zero + plus + minus +
  fixes RepNat :: 'a  $\Rightarrow$  nat
  fixes AbsNat :: nat  $\Rightarrow$  'a
  assumes [simp]: RepNat (AbsNat n) = n
  and [simp]: AbsNat (RepNat x) = x
  and zero-Nat-def: 0 = AbsNat 0
  and plus-Nat-def: a + b = AbsNat (RepNat a + RepNat b)
  and minus-Nat-def: a - b = AbsNat (RepNat a - RepNat b)
begin
  lemma AbsNat-plus: AbsNat (i + j) = AbsNat i + AbsNat j
    <proof>
  lemma AbsNat-zero [simp]: AbsNat 0 + i = i
    <proof>

  subclass comm-monoid-diff
    <proof>
end

```

The type natural numbers is an instantiation of the class *nat*.

```

instantiation nat :: nat
begin

```

```

definition RepNat-nat-def [simp]: (RepNat:: nat  $\Rightarrow$  nat) = id
definition AbsNat-nat-def [simp]: (AbsNat:: nat  $\Rightarrow$  nat) = id
instance  $\langle$ proof $\rangle$ 
end

```

Finally, functions from  $'a :: \text{nat}$  to some arbitrary type  $'b$  are instantiated as a trace class.

```

instantiation fun :: (nat, type) trace
begin
  definition at-trace-def [simp]: ((t :: 'a  $\Rightarrow$  'b)[i..]) j = (t (AbsNat i + j))
instance  $\langle$ proof $\rangle$ 
end

```

By putting together all class definitions and instantiations introduced so far, we obtain the temporal class structure for predicates on traces with arbitrary number of parameters.

For example in the next lemma  $r$  and  $r'$  are predicate relations, and the operator always is available for them as a consequence of the above construction.

```

lemma  $(\Box r) \text{ OO } (\Box r') \leq (\Box (r \text{ OO } r'))$ 
   $\langle$ proof $\rangle$ 
end

```

```

theory Refinement imports Main
begin

```

### 3 Monotonic Predicate Transformers

In this section we introduce the basics of refinement calculus [3]. Part of this theory is a reformulation of some definitions from [5], but here they are given for predicates, while [5] uses sets.

```

notation
  bot ( $\perp$ ) and
  top ( $\top$ ) and
  inf (infixl  $\sqcap$  70)
  and sup (infixl  $\sqcup$  65)

```

#### 3.1 Basic predicate transformers

```

definition
  demonic :: ('a  $\Rightarrow$  'b::lattice)  $\Rightarrow$  'b  $\Rightarrow$  'a  $\Rightarrow$  bool ( $[: - :]$  [0] 1000) where
   $[:Q:]$  p s = (Q s  $\leq$  p)

```

```

definition
  assert::'a::semilattice-inf  $\Rightarrow$  'a  $\Rightarrow$  'a ( $\{. - .\}$  [0] 1000) where
   $\{.p.\}$  q  $\equiv$  p  $\sqcap$  q

```

**definition**

*assume*::('a::boolean-algebra) => 'a => 'a ([. - .] [0] 1000) **where**  
 [.p.] q ≡ (¬p ⊔ q)

**definition**

*angelic* :: ('a ⇒ 'b::{semilattice-inf,order-bot}) ⇒ 'b ⇒ 'a ⇒ bool ({: - :} [0] 1000) **where**  
 {:Q:} p s = (Q s ⊓ p ≠ ⊥)

**syntax**

*-assert* :: patterns => logic => logic ((1{.-/ -}))

**translations**

*-assert* (-patterns x xs) P == CONST assert (CONST id (-abs (-pattern x xs) P))  
*-assert* x P == CONST assert (CONST id (-abs x P))

**term** {x,z . P x y.}

**syntax** *-update* :: patterns => patterns => logic => logic (- ~ - . - 0)

**translations**

*-update* (-patterns x xs) (-patterns y ys) t == CONST id (-abs (-pattern x xs) (CONST id (-abs (-pattern y ys) t)))  
*-update* x y t == CONST id (-abs x (CONST id (-abs y t)))

**term** [x ~ z . (P::'a ⇒ 'b ⇒ 'c ⇒ 'd ⇒ bool) x y y' z :]

**term** [x, y ~ y', z . (P::'a ⇒ 'b ⇒ 'c ⇒ 'd ⇒ bool) x y y' z :]

**term** {: x, y ~ y', z . (P::'a ⇒ 'b ⇒ 'c ⇒ 'd ⇒ bool) x y y' z :}

**lemma** *demonic-demonic*: [:r:] o [:r':] = [:r OO r':]

⟨proof⟩

**lemma** *assert-demonic-comp*: {.p.} o [:r:] o {.p'.} o [:r':] =

{x . p x ∧ (∀ y . r x y → p' y).} o [:r OO r':]

⟨proof⟩

**lemma** *demonic-assert-comp*: [:r:] o {.p.} = {x.(∀ y . r x y → p y).} o [:r:]

⟨proof⟩

**lemma** *assert-assert-comp*: {p::'a::lattice.} o {p'.} = {p ⊓ p'.}

⟨proof⟩

**definition** *inpt* r x = (∃ y . r x y)

**definition** *trs* r = { . inpt r. } o [:r:]

**lemma** *trs* (λ x y . x = y) q x = q x

⟨proof⟩

**lemma** *assert-demonic-prop*:  $\{.p.\} o [:r:] = \{.p.\} o [:(\lambda x y . p x) \sqcap r:]$   
 $\langle proof \rangle$

**lemma** *relcompp-exists*:  $relcompp r r' x y = (\exists u . r x u \wedge r' u y)$   
 $\langle proof \rangle$

**lemma** *trs-trs*:  $(trs r) o (trs r') = trs ((\lambda s t . (\forall s' . r s s' \longrightarrow (inpt r' s')) \sqcap (r OO r'))$  (is ?S = ?T)  
 $\langle proof \rangle$

**lemma** *assert-demonic-refinement*:  $(\{.p.\} o [:r:] \leq \{.p'.\} o [:r':]) = (p \leq p' \wedge (\forall x . p x \longrightarrow r' x \leq r x))$   
 $\langle proof \rangle$

**lemma** *trs-refinement*:  $(trs r \leq trs r') = ((\forall x . inpt r x \longrightarrow inpt r' x) \wedge (\forall x . inpt r x \longrightarrow r' x \leq r x))$   
 $\langle proof \rangle$

**lemma** *trs*  $(\lambda x y . x \geq 0) \leq trs (\lambda x y . x = y)$   
 $\langle proof \rangle$

**lemma** *trs*  $(\lambda x y . x \geq 0) q x = (if q = \top$  then  $x \geq 0$  else *False*)  
 $\langle proof \rangle$

**lemma**  $[:r:] \sqcap [:r':] = [:r \sqcup r':]$   
 $\langle proof \rangle$

**lemma** *spec-demonic-choice*:  $(\{.p.\} o [:r:]) \sqcap (\{.p'.\} o [:r':]) = (\{.p \sqcap p'.\} o [:r \sqcup r':])$   
 $\langle proof \rangle$

**lemma** *trs-demonic-choice*:  $trs r \sqcap trs r' = trs ((\lambda x y . inpt r x \wedge inpt r' x) \sqcap (r \sqcup r'))$   
 $\langle proof \rangle$

**lemma**  $p \sqcap p' = \perp \implies (\{.p.\} o [:r:]) \sqcup (\{.p'.\} o [:r':]) = \{.p \sqcup p'.\} o [:(\lambda x y . p x \longrightarrow r x y) \sqcap ((\lambda x y . p' x \longrightarrow r' x y)):]$   
 $\langle proof \rangle$

### 3.2 Conjunctive predicate transformers

**definition** *conjunctive*  $(S::'a::complete-lattice \Rightarrow 'b::complete-lattice) = (\forall Q . S (Inf Q) = INFIMUM Q S)$

**definition** *sconjunctive*  $(S::'a::complete-lattice \Rightarrow 'b::complete-lattice) = (\forall Q . (\exists x . x \in Q) \longrightarrow S (Inf Q) = INFIMUM Q S)$

**lemma** [*simp*]: *conjunctive*  $S \implies$  *sconjunctive*  $S$   
 $\langle proof \rangle$



**lemma**  $[simp]: conjunctive \top$   
 $\langle proof \rangle$

**lemma**  $conjunctive-demonic [simp]: conjunctive [r:]$   
 $\langle proof \rangle$

**lemma**  $sconjunctive-assert [simp]: sconjunctive \{.p.\}$   
 $\langle proof \rangle$

**lemma**  $sconjunctive-simp: x \in Q \implies sconjunctive S \implies S (Inf Q) = INFIMUM Q S$   
 $\langle proof \rangle$

**lemma**  $sconjunctive-INF-simp: x \in X \implies sconjunctive S \implies S (INFIMUM X Q) = INFIMUM (Q'X) S$   
 $\langle proof \rangle$

**lemma**  $demonic-comp [simp]: sconjunctive S \implies sconjunctive S' \implies sconjunctive (S o S')$   
 $\langle proof \rangle$

**lemma**  $[simp]: conjunctive S \implies S (INFIMUM X Q) = (INFIMUM X (S o Q))$   
 $\langle proof \rangle$

**lemma**  $conjunctive-simp: conjunctive S \implies S (Inf Q) = INFIMUM Q S$   
 $\langle proof \rangle$

**lemma**  $conjunctive-monotonic: sconjunctive S \implies mono S$   
 $\langle proof \rangle$

**definition**  $grd S = - S \perp$

**lemma**  $grd [r:] = inpt r$   
 $\langle proof \rangle$

**definition**  $fail S = -(S \top)$

**definition**  $term S = (S \top)$

**lemma**  $fail (\{.p.\} o [r :: 'a \Rightarrow 'b \Rightarrow bool:]) = -p$   
 $\langle proof \rangle$

**definition**  $Fail = \perp$

**lemma**  $mono (S :: 'a :: boolean-algebra \Rightarrow 'b :: boolean-algebra) \implies (S = Fail) = (fail S = \top)$   
 $\langle proof \rangle$

**definition**  $Skip = id$

**lemma** [simp]:  $\{\cdot, \top :: 'a :: \text{bounded-lattice}.\} = \text{Skip}$   
 ⟨proof⟩

**lemma** [simp]:  $\text{Skip} \circ S = S$   
 ⟨proof⟩

**lemma** [simp]:  $S \circ \text{Skip} = S$   
 ⟨proof⟩

**lemma** [simp]:  $\text{mono } S \implies \text{mono } S' \implies \text{mono } (S \circ S')$   
 ⟨proof⟩

**lemma** [simp]:  $\text{mono } \{\cdot, p :: ('a \Rightarrow \text{bool}).\}$   
 ⟨proof⟩

**lemma** [simp]:  $\text{mono } [::r :: ('a \Rightarrow 'b \Rightarrow \text{bool}):]$   
 ⟨proof⟩

**lemma** [simp]:  $\{\cdot, \lambda x . \text{True} .\} = \text{Skip}$   
 ⟨proof⟩

**lemma** [simp]:  $\perp \circ S = \perp$   
 ⟨proof⟩

**lemma** [simp]:  $\{\cdot, \perp :: 'a :: \text{boolean-algebra}.\} = \perp$   
 ⟨proof⟩

**lemma** [simp]:  $\top \circ S = \top$   
 ⟨proof⟩

**lemma** left-comp:  $T \circ U = T' \circ U' \implies S \circ T \circ U = S \circ T' \circ U'$   
 ⟨proof⟩

**lemma** assert-demonic:  $\{\cdot, p.\} \circ [::r ::] = \{\cdot, p.\} \circ [::\lambda x y . p x \wedge r x y ::]$   
 ⟨proof⟩

**lemma** trs r  $\sqcap$  trs r' = trs  $(\lambda x y . \text{inpt } r x \wedge \text{inpt } r' x \wedge (r x y \vee r' x y))$   
 ⟨proof⟩

### 3.3 Fusion of predicate transformers

In this section we define the fusion operator from [2]. The fusion of two programs  $S$  and  $T$  is intuitively equivalent with the parallel execution of the two programs. If  $S$  and  $T$  assign nondeterministically some value to some program variable  $x$ , then the fusion of  $S$  and  $T$  will assign a value to  $x$  which can be assigned by both  $S$  and  $T$ .

**definition** fusion ::  $(('a \Rightarrow \text{bool}) \Rightarrow ('b \Rightarrow \text{bool})) \Rightarrow (( 'a \Rightarrow \text{bool}) \Rightarrow ('b \Rightarrow \text{bool})) \Rightarrow (( 'a \Rightarrow \text{bool}) \Rightarrow ('b \Rightarrow \text{bool}))$  (**infixl** || 70) **where**

$$(S \parallel S') q x = (\exists (p::'a \Rightarrow \text{bool}) p' . p \sqcap p' \leq q \wedge S p x \wedge S' p' x)$$

**lemma** *fusion-spec*:  $(\{.p.\} \circ [r:]) \parallel (\{.p'.\} \circ [r':]) = (\{.p \sqcap p'.\} \circ [r \sqcap r':])$   
 $\langle \text{proof} \rangle$

**lemma** *fusion-assoc*:  $S \parallel (T \parallel U) = (S \parallel T) \parallel U$   
 $\langle \text{proof} \rangle$

**lemma**  $P \leq Q \Longrightarrow P' \leq Q' \Longrightarrow P \parallel P' \leq Q \parallel Q'$   
 $\langle \text{proof} \rangle$

**lemma** *conjunctive*  $S \Longrightarrow S \parallel \top = \top$   
 $\langle \text{proof} \rangle$

**lemma** *fusion-spec-local*:  $a \in \text{init} \Longrightarrow ([x \rightsquigarrow u, y . u \in \text{init} \wedge x = y:] \circ \{.p.\} \circ [r:]) \parallel (\{.p'.\} \circ [r':])$   
 $= [x \rightsquigarrow u, y . u \in \text{init} \wedge x = y:] \circ \{.u, x . p(u, x) \wedge p' x.\} \circ [u, x \rightsquigarrow y . r(u, x) y \wedge r' x y:]$  (**is**  $?p \Longrightarrow ?S = ?T$ )  
 $\langle \text{proof} \rangle$

**lemma** *fusion-spec-local-a*:  $a \in \text{init} \Longrightarrow ([x \rightsquigarrow u, y . u \in \text{init} \wedge x = y:] \circ \{.p.\} \circ [r:]) \parallel [r':]$   
 $= ([x \rightsquigarrow u, y . u \in \text{init} \wedge x = y:] \circ \{.p.\} \circ [u, x \rightsquigarrow y . r(u, x) y \wedge r' x y:])$   
 $\langle \text{proof} \rangle$

**lemma** *fusion-local-refinement*:

$a \in \text{init} \Longrightarrow (\bigwedge x u y . u \in \text{init} \Longrightarrow p' x \Longrightarrow r(u, x) y \Longrightarrow r' x y) \Longrightarrow$   
 $\{.p'.\} \circ ([x \rightsquigarrow u, y . u \in \text{init} \wedge x = y:] \circ \{.p.\} \circ [r:]) \parallel [r':] \leq [x \rightsquigarrow u,$   
 $y . u \in \text{init} \wedge x = y:] \circ \{.p.\} \circ [r:]$   
 $\langle \text{proof} \rangle$

**end**

**theory** *Reactive*

**imports** *Temporal Refinement*

**begin**

## 4 Reactive Systems

In this section we introduce reactive systems which are modeled as monotonic property transformers where properties are predicates on traces. We start with introducing some examples that uses LTL to specify global behaviour on traces, and later we introduce property transformers based on symbolic transition systems.

**definition** *HAVOC* =  $[x \rightsquigarrow y . \text{True}]$

**definition** *ASSERT-LIVE* =  $\{. \square \diamond (\lambda x . x \theta) .\}$

**definition** *GUARANTY-LIVE* =  $[x \rightsquigarrow y . \square \diamond (\lambda y . y \theta)]$

**definition** *AE* = *ASSERT-LIVE*  $\circ$  *HAVOC*

**definition** *SKIP* =  $[x \rightsquigarrow y . x = y]$

**lemma** *[simp]*:  $SKIP = id$   
 $\langle proof \rangle$

**definition**  $REQ-RESP = [ : \square(\lambda xs\ ys . xs\ (0::nat) \longrightarrow (\diamond (\lambda ys . ys\ (0::nat))))$   
 $ys) : ]$

**definition**  $FAIL = \perp$

**lemma**  $HAVOC\ o\ ASSERT-LIVE = FAIL$   
 $\langle proof \rangle$

**lemma**  $HAVOC\ o\ AE = FAIL$   
 $\langle proof \rangle$

**lemma**  $HAVOC\ o\ ASSERT-LIVE = FAIL$   
 $\langle proof \rangle$

**lemma**  $SKIP\ o\ AE = AE$   
 $\langle proof \rangle$

**lemma**  $(REQ-RESP\ o\ AE) = AE$   
 $\langle proof \rangle$

## 4.1 Symbolic transition systems

In this section we introduce property transformers basend on symbolic transition systems. These are systems with local state. The execution starts in some initial state, and with some input value the system computes a new state and an output value. Then using the current state, and a new input value the system computes a new state, and a new output, and so on. The system may fail if at some point the input and the current state do not satisfy a required predicate.

In the folowing definitions the variables  $u$ ,  $x$ ,  $y$  stand for the state of the system, the input, and the output respectively. The *init* is the property that the initial state should satisfy. The predicate  $p$  is the precondition of the input and the current state, and the relation  $r$  gives the next state and the output based on the input and the current state.

**definition**  $fail-sys\ init\ p\ r\ x = (\exists n\ u\ y . u \in init \wedge (\forall i < n . r\ (u\ i)\ (u\ (Suc\ i))\ (x\ i)\ (y\ i)) \wedge (\neg p\ (u\ n)\ (u\ (Suc\ n))\ (x\ n)))$

**definition**  $run\ r\ u\ x\ y = (\forall i . r\ (u\ i)\ (u\ (Suc\ i))\ (x\ i)\ (y\ i))$

**definition**  $LocalSystem\ init\ p\ r\ q\ x = (\neg fail-sys\ init\ p\ r\ x \wedge (\forall u\ y . (u \in init \wedge run\ r\ u\ x\ y) \longrightarrow q\ y))$

**lemma**  $fail\ (LocalSystem\ init\ p\ r) = fail-sys\ init\ p\ r$   
 $\langle proof \rangle$

**definition**  $inpt-st\ r\ u\ u'\ x = (\exists y . r\ u\ u'\ x\ y)$

**definition**  $lft\text{-}pred\text{-}st\ p\ u\ x = p\ (u\ (0::nat))\ (u\ 1)\ (x\ (0::nat))$

**definition**  $lft\text{-}pred\text{-}loc\text{-}st\ p\ u\ x = p\ (u\ (0::nat))\ (x\ (0::nat))$

**definition**  $lft\text{-}rel\text{-}st\ r\ u\ x\ y = r\ (u\ (0::nat))\ (u\ 1)\ (x\ (0::nat))\ (y\ (0::nat))$

**definition**  $prec\text{-}st\ p\ r = \neg((lft\text{-}pred\text{-}st\ (inpt\text{-}st\ r))\ until\ \neg(lft\text{-}pred\text{-}st\ p))$

**lemma**  $prec\text{-}st\text{-}simp$ :  $prec\text{-}st\ p\ r\ u\ x = (\forall\ n . (\forall\ i < n . inpt\text{-}st\ r\ (u\ i)\ (u\ (Suc\ i))\ (x\ i)) \longrightarrow p\ (u\ n)\ (u\ (Suc\ n))\ (x\ n))$   
 $\langle proof \rangle$

**definition**  $SymSystem\ init\ p\ r = [z \rightsquigarrow u, x . u \in init \wedge z = x:] \circ \{.u, x . prec\text{-}st\ p\ r\ u\ x.\} \circ [ :u, x \rightsquigarrow y . (\Box\ (lft\text{-}rel\text{-}st\ r))\ u\ x\ y :]$

**lemma**  $SymSystem\text{-}rel$ :  $SymSystem\ init\ p\ r = \{. x . \forall u . u \in init \longrightarrow prec\text{-}st\ p\ r\ u\ x .\} \circ [ :x \rightsquigarrow y . \exists u . u \in init \wedge (\Box\ lft\text{-}rel\text{-}st\ r)\ u\ x\ y :]$   
 $\langle proof \rangle$

**theorem**  $SymSystem\ init\ p\ r\ q\ x = LocalSystem\ init\ p\ r\ q\ x$   
 $\langle proof \rangle$

**definition**  $local\text{-}init\ init\ S = INFIMUM\ init\ S$

**definition**  $zip\text{-}set\ A\ B = \{u . ((fst\ o\ u) \in A) \wedge ((snd\ o\ u) \in B)\}$

**definition**  $nzip:: ('x \Rightarrow 'a) \Rightarrow ('x \Rightarrow 'b) \Rightarrow 'x \Rightarrow ('a \times 'b)$  (**infixl** || 65) **where**  $(xs\ ||\ ys)\ i = (xs\ i, ys\ i)$

**lemma**  $[simp]$ :  $fst\ o\ x\ ||\ y = x$   
 $\langle proof \rangle$

**lemma**  $[simp]$ :  $snd\ o\ x\ ||\ y = y$   
 $\langle proof \rangle$

**lemma**  $[simp]$ :  $x \in A \Longrightarrow y \in B \Longrightarrow (x\ ||\ y) \in zip\text{-}set\ A\ B$   
 $\langle proof \rangle$

**lemma**  $local\text{-}demonic\text{-}init$ :  $local\text{-}init\ init\ (\lambda\ u . \{. x . p\ u\ x.\} \circ [ :x \rightsquigarrow y . r\ u\ x\ y :]) = [ :z \rightsquigarrow u, x . u \in init \wedge z = x:] \circ \{.u, x . p\ u\ x.\} \circ [ :u, x \rightsquigarrow y . r\ u\ x\ y :]$   
 $\langle proof \rangle$

**lemma**  $local\text{-}init\text{-}comp$ :  $u' \in init' \Longrightarrow (\forall\ u . sconjunctive\ (S\ u)) \Longrightarrow (local\text{-}init\ init\ S) \circ (local\text{-}init\ init'\ S') = local\text{-}init\ (zip\text{-}set\ init\ init')\ (\lambda\ u . (S\ (fst\ o\ u)) \circ (S'\ (snd\ o\ u)))$

$\langle \text{proof} \rangle$

**lemma** *init-state*:  $[z \rightsquigarrow u, x . u \in \text{init} \wedge z = x:] \circ \{.u, x . p \ u \ x.\} \circ [u, x \rightsquigarrow y . r \ u \ x \ y :]$   
 $= [z \rightsquigarrow u, x . z = x:] \circ \{.u, x . u \in \text{init} \longrightarrow p \ u \ x.\} \circ [u, x \rightsquigarrow y . u \in \text{init} \wedge r \ u \ x \ y :]$   
 $\langle \text{proof} \rangle$

**lemma** *always-lft-rel-comp*:  $(\Box \text{ lft-rel-st } r) (fst \circ u) \text{ OO } (\Box \text{ lft-rel-st } r') (snd \circ u)$   
 $= (\Box \text{ lft-rel-st } (\lambda (u, v) (u', v') . ((r \ u \ u') \text{ OO } (r' \ v \ v')))) \ u$   
 $\langle \text{proof} \rangle$

**theorem** *SymSystem-comp*:  $u' \in \text{init}' \implies \text{SymSystem } \text{init } p \ r \ o \ \text{SymSystem } \text{init}' \ p' \ r'$   
 $= [z \rightsquigarrow u, x . fst \ o \ u \in \text{init} \wedge snd \ o \ u \in \text{init}' \wedge z = x:]$   
 $\circ \{. u, x . \text{prec-st } p \ r \ (fst \circ u) \ x \wedge (\forall y. (\Box \text{ lft-rel-st } r) (fst \circ u) \ x$   
 $y \longrightarrow \text{prec-st } p' \ r' \ (snd \circ u) \ y) .\}$   
 $\circ [u, x \rightsquigarrow y . (\Box \text{ lft-rel-st } (\lambda(u, v) (u', v') . r \ u \ u' \text{ OO } r' \ v \ v')) \ u$   
 $x \ y :]$   
 $(\text{is } ?p \implies ?S = ?T)$   
 $\langle \text{proof} \rangle$

**lemma** *always-lft-rel-comp-a*:  $(\Box \text{ lft-rel-st } r) \ u \text{ OO } (\Box \text{ lft-rel-st } r') \ v$   
 $= (\Box \text{ lft-rel-st } (\lambda (u, v) (u', v') . ((r \ u \ u') \text{ OO } (r' \ v \ v')))) (u \parallel v)$   
 $\langle \text{proof} \rangle$

**theorem** *SymSystem-comp-a*:  $u' \in \text{init}' \implies \text{SymSystem } \text{init } p \ r \ o \ \text{SymSystem } \text{init}' \ p' \ r'$   
 $= \{.x . \forall u \ v . u \in \text{init} \wedge v \in \text{init}' \longrightarrow (\text{prec-st } p \ r \ u \ x \wedge (\forall y. (\Box$   
 $\text{lft-rel-st } r) \ u \ x \ y \longrightarrow \text{prec-st } p' \ r' \ v \ y)) .\}$   
 $\circ [u, x \rightsquigarrow y . \exists u \ v . u \in \text{init} \wedge v \in \text{init}' \wedge (\Box \text{ lft-rel-st } (\lambda(u, v)$   
 $(u', v') . r \ u \ u' \text{ OO } r' \ v \ v')) (u \parallel v) \ x \ y :]$   
 $(\text{is } ?p \implies ?S = ?T)$   
 $\langle \text{proof} \rangle$

We show next that the composition of two SymSystem  $S$  and  $S'$  is not equal to the SymSystem of the composition of local transitions of  $S$  and  $S'$

**definition**  $\text{init}S = \{u . fst (u \ (0::nat)) = (0::nat)\}$

**definition**  $\text{localPrec}S = (\top:: nat \times nat \Rightarrow nat \times nat \Rightarrow nat \Rightarrow bool)$

**definition**  $\text{localRel}S = (\lambda (u::nat, v) (u', v'::nat) (x::nat) (y::nat) . u = 0 \wedge u' = 1 \wedge v = v')$

**definition**  $\text{init}S' = (\top::(nat \Rightarrow (nat \times nat)) \text{ set})$

**definition**  $\text{localPrec}S' = (\perp:: nat \times nat \Rightarrow nat \times nat \Rightarrow nat \Rightarrow bool)$

**definition**  $\text{localRel}S' = (\lambda (u::nat, v) (u', v'::nat) (x::nat) (y::nat) . u = u')$

**definition**  $\text{symb}S = \text{SymSystem } \text{init}S \ \text{localPrec}S \ \text{localRel}S$

**definition**  $\text{symbS}' = \text{SymSystem } \text{initS}' \text{ localPrecS}' \text{ localRelS}'$

**definition**  $\text{localPrecSS}' = (\lambda(u::\text{nat}, v::\text{nat}) (u', v') (x::\text{nat}) . 0 < u)$

**definition**  $\text{localRelSS}' = (\lambda (u, v::\text{nat}) (u'::\text{nat}, v'::\text{nat}) (x::\text{nat}) (z::\text{nat}) . (u::\text{nat}) = 0 \wedge u' = 1)$

**lemma**  $\text{localSS}'\text{-aux}$ :  $(\lambda x. \forall (a::\text{nat}) (aa::\text{nat}) (b::\text{nat}). \neg (\text{case } x \text{ of } (x::\text{nat}, u::\text{nat}, v::\text{nat}) \Rightarrow \lambda ab. u = 0 \wedge (\text{case } ab \text{ of } (y, u', v') \Rightarrow u' = \text{Suc } 0 \wedge v = v') (a, aa, b)) = (\lambda (x, u, v) . u > 0)$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{localSS}'\text{-aux-b}$ :  $((\lambda(x, u, v) ab. u = 0 \wedge (\text{case } ab \text{ of } (y, u', v') \Rightarrow u' = \text{Suc } 0 \wedge v = v')) \text{ OO } (\lambda(x, u, v) (y, u', v'). u = u')) = (\lambda (x, u, v) (y, u', v') . u = 0 \wedge u' = 1)$   
 $\langle \text{proof} \rangle$

**lemma**  $\{.x, (u, v) . \text{localPrecS}' (u, v) (a, b) x.\} \circ [ :x, (u, v) \rightsquigarrow y, (u', v') . \text{localRelS}' (u, v) (u', v') x y:] \circ \{.x, (u, v) . \text{localPrecS}' (u, v) (c, d) x.\} \circ [ :x, (u, v) \rightsquigarrow y, (u', v') . \text{localRelS}' (u, v) (u', v') x y:] = \{.x, (u, v) . \text{localPrecSS}' (u, v) (e, f) x.\} \circ [ :x, (u, v) \rightsquigarrow y, (u', v') . \text{localRelSS}' (u, v) (u', v') x y:]$   
 $\langle \text{proof} \rangle$

**lemma**  $[\text{simp}]$ :  $[ : \perp :: ('a \Rightarrow 'b \Rightarrow ('c::\text{boolean-algebra})) : ] = \top$   
 $\langle \text{proof} \rangle$

**definition**  $\text{symbSS}' = \text{SymSystem } \text{initS}' \text{ localPrecSS}' \text{ localRelSS}'$

**lemma**  $\text{symbSS}'\text{-aux}$ :  $(\lambda x. \forall u. \text{fst } (u \ 0) = 0 \longrightarrow (\forall n. (\forall i < n. \text{Ex } ((\text{case } u \ i \text{ of } (u, v) \Rightarrow \lambda(u', v'::\text{nat}) x z. u = 0 \wedge u' = \text{Suc } 0) (u (\text{Suc } i)) (x \ i)))) \longrightarrow (\text{case } u \ n \text{ of } (u, v) \Rightarrow \lambda(u', v') x. 0 < u) (u (\text{Suc } n)) (x \ n))) = \perp$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{symbSS}'$ :  $\text{symbSS}' = \perp$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{symbS}$ :  $\text{symbS} = \top$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{symbS} \circ \text{symbS}' \neq \text{symbSS}'$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{prec-st-inpt}$ :  $\text{prec-st } (\text{inpt-st } r) r = (\sqcap (\text{lft-pred-st } (\text{inpt-st } r)))$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{grd } (\text{SymSystem } \text{init } p \ r) = \text{SUPREMUM } \text{init } (-\text{prec-st } p \ r \sqcup (\square (\text{lft-pred-st } (\text{inpt-st } r))))$   
 $\langle \text{proof} \rangle$

**definition**  $\text{guard } S = \{.((\text{grd } S)::'a \Rightarrow \text{bool}).\} \circ S$

**lemma**  $((\text{grd } (\text{local-init } \text{init } S))::'a \Rightarrow \text{bool}) = \text{SUPREMUM } \text{init } (\text{grd } \circ S)$   
 $\langle \text{proof} \rangle$

**lemma**  $u \in \text{init} \Rightarrow \text{guard } ([:z \rightsquigarrow u, x . u \in \text{init} \wedge z = x:] \circ \{.u, x . p \ u \ x.\} \circ [:u, x \rightsquigarrow y . r \ u \ x \ y :])$   
 $= [:z \rightsquigarrow u, x . u \in \text{init} \wedge z = x:] \circ \{.u, x . u \in \text{init} \wedge (\exists a. a \in \text{init} \wedge (p \ a \ x \rightarrow \text{Ex } (r \ a \ x))) \wedge p \ u \ x.\} \circ [:u, x \rightsquigarrow y . ((r \ u \ x \ y)::\text{bool}) :]$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{inpt-str-comp-aux}: (\forall n. (\forall i < n. \text{inpt-st } (\lambda(u, v) (u', v'). r \ u \ u' \ O O \ r' \ v \ v') (u \ i) (u \ (\text{Suc } i)) (x \ i)) \rightarrow \text{inpt-st } r \ (\text{fst } (u \ n)) (\text{fst } (u \ (\text{Suc } n))) (x \ n) \wedge (\forall y. r \ (\text{fst } (u \ n)) (\text{fst } (u \ (\text{Suc } n))) (x \ n) \ y) \rightarrow \text{inpt-st } r' \ (\text{snd } (u \ n)) (\text{snd } (u \ (\text{Suc } n))) (y)) \rightarrow (\forall i < n. \text{inpt-st } r \ ((\text{fst } \circ u) \ i) ((\text{fst } \circ u) \ (\text{Suc } i)) (x \ i) \wedge (\forall y. r \ (\text{fst } (u \ i)) (\text{fst } (u \ (\text{Suc } i))) (x \ i) \ y) \rightarrow \text{inpt-st } r' \ (\text{snd } (u \ i)) (\text{snd } (u \ (\text{Suc } i))) (y))$   
 $(\text{is } (\forall n . ?p \ n) \rightarrow ?q \ n)$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{inpt-str-comp-aux-a}: (\forall n. (\forall i < n. \text{inpt-st } (\lambda(u, v) (u', v'). r \ u \ u' \ O O \ r' \ v \ v') (u \ i) (u \ (\text{Suc } i)) (x \ i)) \rightarrow \text{inpt-st } r \ (\text{fst } (u \ n)) (\text{fst } (u \ (\text{Suc } n))) (x \ n) \wedge (\forall y. r \ (\text{fst } (u \ n)) (\text{fst } (u \ (\text{Suc } n))) (x \ n) \ y) \rightarrow \text{inpt-st } r' \ (\text{snd } (u \ n)) (\text{snd } (u \ (\text{Suc } n))) (y)) \Rightarrow \text{inpt-st } r \ ((\text{fst } \circ u) \ n) ((\text{fst } \circ u) \ (\text{Suc } n)) (x \ n) \wedge (\forall y. r \ (\text{fst } (u \ n)) (\text{fst } (u \ (\text{Suc } n))) (x \ n) \ y) \rightarrow \text{inpt-st } r' \ (\text{snd } (u \ n)) (\text{snd } (u \ (\text{Suc } n))) (y))$   
 $\langle \text{proof} \rangle$

**definition**  $\text{rel-st } r \ r' = (\lambda (u, v) (u', v') \ x \ z . \text{inpt-st } r \ u \ u' \ x \wedge (\forall y . r \ u \ u' \ x \ y \rightarrow \text{inpt-st } r' \ v \ v' \ y) \wedge (r \ u \ u' \ O O \ r' \ v \ v') \ x \ z)$

**lemma**  $\text{inpt-str-comp-a}: (\text{prec-st } (\text{inpt-st } r) \ r \ (\text{fst } \circ u) \ x \wedge (\forall y. (\square \ \text{lft-rel-st } r) (\text{fst } \circ u) \ x \ y \rightarrow \text{prec-st } (\text{inpt-st } r') \ r' \ (\text{snd } \circ u) \ y)) = \text{prec-st } (\lambda \ u \ u' \ x . \text{inpt-st } r \ (\text{fst } u) (\text{fst } u') \ x \wedge (\forall y . r \ (\text{fst } u) (\text{fst } u') \ x \ y \rightarrow (\text{inpt-st } r' \ (\text{snd } u) (\text{snd } u') \ y))) (\lambda(u, v) (u', v'). r \ u \ u' \ O O \ r' \ v \ v') \ u \ x$   
 $\langle \text{proof} \rangle$

**lemma**  $\text{inpt-str-comp-b}: \text{prec-st } (\lambda \ u \ u' \ x . \text{inpt-st } r \ (\text{fst } u) (\text{fst } u') \ x \wedge (\forall y . r \ (\text{fst } u) (\text{fst } u') \ x \ y \rightarrow (\text{inpt-st } r' \ (\text{snd } u) (\text{snd } u') \ y))) (\lambda(u, v) (u', v'). r \ u \ u' \ O O \ r' \ v \ v') \ u \ x$   
 $= (\square (\text{lft-pred-st } (\text{inpt-st } (\text{rel-st } r \ r')))) \ u \ x$   
 $\langle \text{proof} \rangle$



**lemma** *inpt-str-comp*:  $(\text{prec-st } (\text{inpt-st } r) r (\text{fst } \circ u) x \wedge (\forall y. (\Box \text{lft-rel-st } r) (\text{fst } \circ u) x y \longrightarrow \text{prec-st } (\text{inpt-st } r') r' (\text{snd } \circ u) y))$   
 $= (\Box (\text{lft-pred-st } (\text{inpt-st } (\text{rel-st } r r')))) u x$   
 ⟨proof⟩

**lemma** *RSysTmp-inpt-comp*:  $u' \in \text{init}' \implies \text{SymSystem } \text{init } (\text{inpt-st } r) r \circ \text{SymSystem } \text{init}' (\text{inpt-st } r') r'$   
 $= \text{SymSystem } (\text{zip-set } \text{init } \text{init}') (\text{inpt-st } (\text{rel-st } r r')) (\text{rel-st } r r')$   
 ⟨proof⟩

**definition** *GrdSymSystem*  $\text{init } r = [z \rightsquigarrow u, x . u \in \text{init} \wedge z = x:] \circ \text{trs } (\lambda (u, x) y . (\Box (\text{lft-rel-st } r)) u x y)$

**lemma** *inpt-always*:  $\text{inpt } (\lambda(x, y). (\Box \text{lft-rel-st } r) x y) = (\lambda(x, y). (\Box \text{lft-pred-st } (\text{inpt-st } r)) x y)$   
 ⟨proof⟩

**lemma** *GrdSymSystem*  $\text{init } r = \text{SymSystem } \text{init } (\text{inpt-st } r) r$   
 ⟨proof⟩

## 4.2 Example: COUNTER

In this section we introduce an example counter that counts how many times the input variable  $x$  is true. The input is a sequence of boolean values and the output is a sequence of natural numbers. The output at some moment in time is the number of true values seen so far in the input.

We defined the system counter in two different ways and we show that the two definitions are equivalent. The first definition takes the entire input sequence and it computes the corresponding output sequence. We introduce the second version of the counter as a reactive system based on a symbolic transition system. We use a local variable to record the number of true values seen so far, and initially the local variable is zero. At every step we increase the local variable if the input is true. The output of the system at every step is equal to the local variable.

**primrec** *count* ::  $\text{bool trace} \Rightarrow \text{nat trace}$  **where**  
 $\text{count } x \ 0 = (\text{if } x \ 0 \text{ then } 1 \text{ else } 0) \mid$   
 $\text{count } x \ (\text{Suc } n) = (\text{if } x \ (\text{Suc } n) \text{ then } \text{count } x \ n + 1 \text{ else } \text{count } x \ n)$

**definition** *Counter-global*  $n = \{.x . (\forall k . \text{count } x \ k \leq n).\} \circ [x \rightsquigarrow y . y = \text{count } x:]$

**definition** *prec-count*  $M \ u \ u' \ x = (u \leq M)$

**definition** *rel-count*  $u \ u' \ x \ y = ((x \longrightarrow u' = \text{Suc } u) \wedge (\neg x \longrightarrow u' = u) \wedge y = u')$

**lemma** *counter-a-ax*:  $u \ 0 = 0 \implies \forall i < n. (x \ i \longrightarrow u \ (\text{Suc } i) = \text{Suc } (u \ i)) \wedge (\neg x \ i \longrightarrow u \ (\text{Suc } i) = u \ i) \implies (\forall i < n . \text{count } x \ i = u \ (\text{Suc } i))$

$\langle proof \rangle$

**lemma** *counter-b-aux*:  $u \ 0 = 0 \implies \forall n. (xa \ n \longrightarrow u \ (Suc \ n) = Suc \ (u \ n)) \wedge$   
 $(\neg \ xa \ n \longrightarrow u \ (Suc \ n) = u \ n) \wedge \ xb \ n = u \ (Suc \ n)$   
 $\implies \ count \ xa \ n = u \ (Suc \ n)$

$\langle proof \rangle$

**definition** *COUNTER*  $M = SymSystem \ \{u \ . \ u \ 0 = 0\}$  (*prec-count*  $M$ ) *rel-count*

**lemma** *COUNTER* = *Counter-global*

$\langle proof \rangle$

### 4.3 Example: LIVE

The last example of this formalization introduces a system which does some local computation, and ensures some global liveness property. We show that this example is the fusion of a symbolic transition system and a demonic choice which ensures the liveness property of the output sequence. We also show that assuming some liveness property for the input, we can refine the example into an executable system that does not ensure the liveness property of the output on its own, but relies on the liveness of the input.

**definition** *rel-ex*  $u \ u' \ x \ y = (((x \wedge u' = u + (1::int)) \vee (\neg \ x \wedge u' = u - 1) \vee u' = 0) \wedge (y = (u' = 0)))$

**definition** *prec-ex*  $u \ u' \ x = (-1 \leq u \wedge u \leq 3)$

**definition** *LIVE* =  $[:x \rightsquigarrow u, x' \ . \ u \ (0::nat) = 0 \wedge x = x':] \ o \ \{.u, x \ . \ prec\text{-}st$   
*prec-ex rel-ex*  $u \ x.\}$

$\ o \ [ :u, x \rightsquigarrow y \ . \ (\Box (\lambda \ u \ x \ y \ . \ rel\text{-}ex \ (u \ 0) \ (u \ 1) \ (x \ 0) \ (y \ 0))) \ u \ x \ y \wedge (\Box (\Diamond (\lambda \ y \ . \ y \ 0))) \ y :]$

**thm** *fusion-spec-local-a*

**lemma** *LIVE-fusion*:  $LIVE = (SymSystem \ \{u \ . \ u \ 0 = 0\} \ prec\text{-}ex \ rel\text{-}ex) \parallel [ :x$   
 $\rightsquigarrow y \ . \ (\Box (\Diamond (\lambda \ y \ . \ y \ 0))) \ y :]$

$\langle proof \rangle$

**definition** *preca-ex*  $x = (x \ 1 = (\neg \ x \ 0))$

**lemma** *monotonic-SymSystem[simp]*: *mono* (*SymSystem* *init*  $p \ r$ )

$\langle proof \rangle$

**lemma** *event-ex-aux-a*:  $a \ 0 = (0::int) \implies \forall n. \ xa \ (Suc \ n) = (\neg \ xa \ n) \implies$   
 $\forall n. (xa \ n \wedge a \ (Suc \ n) = a \ n + 1 \vee \neg \ xa \ n \wedge a \ (Suc \ n) = a \ n - 1 \vee a$   
 $(Suc \ n) = 0) \implies$

$(a \ n = -1 \longrightarrow xa \ n) \wedge (a \ n = 1 \longrightarrow \neg \ xa \ n) \wedge -1 \leq a \ n \wedge a \ n \leq 1$

$\langle proof \rangle$

**lemma** *event-ex-aux*:  $a \ 0 = (0::int) \implies \forall n. \ xa \ (Suc \ n) = (\neg \ xa \ n) \implies$

$$\begin{aligned} & \forall n. (xa\ n \wedge a\ (Suc\ n) = a\ n + 1 \vee \neg xa\ n \wedge a\ (Suc\ n) = a\ n - 1 \vee a \\ & (Suc\ n) = 0) \implies \\ & (\forall n. (a\ n = -1 \longrightarrow xa\ n) \wedge (a\ n = 1 \longrightarrow \neg xa\ n) \wedge -1 \leq a\ n \wedge a\ n \leq \\ & 1) \\ & \langle proof \rangle \end{aligned}$$

**lemma**  $\{\cdot, \square\}$  *preca-ex.*  $\} o\ LIVE \leq SymSystem\ \{u.\ u\ 0 = 0\}$  *prec-ex rel-ex*  
 $\langle proof \rangle$   
**end**

## References

- [1] R.-J. Back. *On the correctness of refinement in program development*. PhD thesis, Department of Computer Science, University of Helsinki, 1978.
- [2] R.-J. Back and M. Butler. Exploring summation and product operators in the refinement calculus. In B. Möller, editor, *Mathematics of Program Construction*, volume 947 of *Lecture Notes in Computer Science*, pages 128–158. Springer Berlin Heidelberg, 1995.
- [3] R.-J. Back and J. von Wright. *Refinement Calculus. A systematic Introduction*. Springer, 1998.
- [4] D. Harel and A. Pnueli. On the development of reactive systems. In K. R. Apt, editor, *Logics and Models of Concurrent Systems*, pages 477–498. 1985.
- [5] V. Preoteasa and R.-J. Back. Semantics and data refinement of invariant based programs. In G. Klein, T. Nipkow, and L. Paulson, editors, *The Archive of Formal Proofs*. <http://isa-afp.org/entries/DataRefinementIBP.shtml>, May 2010. Formal proof development.
- [6] V. Preoteasa and S. Tripakis. Refinement calculus of reactive systems. Jun 2014. <http://arxiv.org/abs/1406.6035>.
- [7] S. Tripakis, B. Lickly, T. A. Henzinger, and E. A. Lee. A theory of synchronous relational interfaces. *ACM Trans. Program. Lang. Syst.*, 33(4):14:1–14:41, July 2011.